
· If several users are competing to access the same resource or resources, you need to find a way of synchronizing the access in order to protect the consistency of your data. 

· Example: In a flight booking system, you would need to check whether seats are still free  You also need a guarantee that critical data (the number of free seats in this case) cannot be changed while you are working with the program.

· Locks are a way of coordinating competing accesses to a resource. Each user requests a lock before accessing critical data. 

· It is important to release the lock as soon as possible so as not to hinder other users unnecessarily.

· The Database Management System (DBMS) physically locks the table lines that are read with the intent of being changed ("SELECT SINGLE <f> FROM <dbtab> FOR UPDATE") (database locks).  Other users who want to access the locked record or records must wait until the physical lock has been released.

· At the end of a database LUW, that is, with every DB commit, the database system releases all the locks set during the DB LUW.

· In SAP R/3, this means that each database lock is released when a new screen is displayed, since a change of screen triggers an implicit database commit.  Therefore, database locks are not sufficient if data is collected throughout several screens and the respective data records are to be kept locked during this time frame.

· To keep a lock set through a series of screens (from the dialog program to the update program), the R/3 System has a global lock table at application server level, which you can use to set logical locks for table entries. 

· The lock table and the respective enqueue work process that manages the lock table are on a uniquely defined application server of the SAP R/3 system. All the logical lock requirements of SAP R/3 – regardless of the application server where they are triggered – run on the basis of this system-wide, unique work process. 

· You can also use logical locks to "lock" table entries that do not yet exist on the database.  This is appropriate, for example, when you enter new table lines, and not possible with the help of database locks.

· For further information, refer to the keyword documentation in the ABAP Editor on locking mechanism.  

· You set a logical lock in the lock table by calling a lock module.   This kind of lock module is a special, table-related function module that is created automatically when you activate a table-related lock object. When you call the lock module, logical locks are set for entries in the respective table(s). 

· You maintain lock objects in the Dictionary.  Customer lock objects must begin with EY or EZ.  When you create a lock object, you only need to specify the table whose entries are to be locked later (primary or basis table).  However, you can specify other tables that have a foreign-key relationship to the primary table (secondard tables). By doing so, you can lock an appropriate combination of table entries that belong together through the automatically created lock module. 
Example: A lock object that contains the tables SFLIGHT (primary) and SBOOK (secondary) enables you to lock a flight with its bookings.

· The lock module that is automatically created by the system contains, as input parameters, the lock parameters that are contained in the lock object (among other things). These lock parameters are used to communicate to the lock module which records are to have a logical lock set in the locking table.   The system automatically proposes as names for the lock parameters those names contained in the table key fields (which can be overwritten).

· Also, when you are defining a lock object, you can define the lock mode (E, X, or S) for each specified table. This lock mode functions as a default entry for the lock module and can be overwritten. The different lock modes are described in detail later in this unit.

· Whenever a lock object is successfully activated, the system automatically generates one function module each for setting and releasing locks for entries in the table(s) specified in the lock object.

· The function modules have the following naming convention:
· ENQUEUE_<Lock-Object-Name> to set locks

· DEQUEUE_<Lock-Object-Name> to release locks

· A logical lock is set when you call the lock module; this writes a corresponding entry into the lock table.

· You can only set a lock if no lock entries exist in the lock table for the respective table records.

· If there is an error, the lock module triggers appropriate exceptions. For this reason, the application program can determine the success or failure of the lock on the basis of the return code delivered by the lock module (see the specifications on the slide), and react accordingly.  If there is an error, for example, the current user receives an error message stating that he or she has been rejected by the system.  

· Depending on the bundle technique used for database updates (see the unit Organizing Database Changes), an application program may need to delete again the lock entries it has created (during inline update) or have them deleted automatically (during the update task).

· If an application program that has created lock entries is terminated, the locks are released automatically (implicitly).  Program termination takes place, for example, if you have messages of the type "A" / "X",  if you have the statements  LEAVE PROGRAM  and  LEAVE TO TRANSACTION ,  or if the user enters  "/n" in the command field..

· Using the import parameters of an ENQUEUE function module that correspond to the key fields in the respective table, you determine the table lines to be blocked, that is, you find out for which table lines a lock entry is to be written to the lock table.  These import parameters are called lock parameters.
· If the system could not set the lock successfully (sy-subrc not 0), you should issue a corresponding dialog message to the current user.

· At the end of the dialog program, you can use the corresponding DEQUEUE function module to delete the entries from the lock table.

· DEQUEUE function modules do not trigger any exceptions.  If you attempt to release an entry that is not locked, this has no effect.

· If you want to release all of the locks that you have set - at the end of your dialog program, for example - you can use the function module DEQUEUE_ALL.

· The core part of a lock entry is the lock argument that consists of lock parameters (key fields of the respective table) and the key of the table entry (entries) to be locked.  .

· If, when you call a lock module, a lock parameter is set to its initial value or not specified at all, the system will interpret this as a generic value, that is, it will interpret the lock as referring to all table lines to which the other parameter variants apply.

· The client parameter is an exception.  The following rules apply when you call the lock module
- If the parameter is not specified, the lock applies only to the current execution client (SY-MANDT)
- If the parameter is specified for a particular client, the lock applies to this client
- If the parameter is specified with SPACE, the lock applies to all clients.

· To display the lock table, use transaction SM12.

· The mode_<TabName>  parameter overrides the default lock mode of the lock module that is specified in the lock object.

· Using the parameter x_<Sperrparameter>  you can lock the table line that has an initial value in the corresponding lock parameter field.    (See comment for slide "Lock argument")

· The parameter _scope defines the validity area of the lock :
 1 :  The lock remains set in the program and is deleted again by the program.
         (Usage for inline updates)
 2 :  The lock is passed to the update program (default)
         (Usage for udpates through update program; see the unit Organizing Database Updates).
 3 :  Two locks are set.  One stays set in the program and is deleted from there.  The other one is passed to the update program.  These functions are required whenever the update task is triggered for a partial update of a record, but the corresponding total record should still remain locked in the program for further partial updates.

· The _wait  parameter defines whether a lock request should be repeated in case the first lock attempt fails. You can configure the retry interval by setting the profile parameter (ENQUEUE/DELAYMAX).

· You use the parameter _collect  to store the lock request in the local lock container until it is collected and passed later on to the enqueue server.   Refer also to the slide Usage of the Lock Container

· Requesting a lock from a program is a communication step with lock administration.  If your program sets locks for several objects, this communication effort occurs more than once.

· If you use the lock container, you can reduce the technical effort required for this step.  To do this, set the parameter  _collect  = 'X' whenever you call the lock module. This has the effect that the respective lock requests are stored in the local container for subsequent collective dispatch. 

· You can send the contents of the lock container afterwards to the lock management using the function module FLUSH_ENQUEUE. 

· If you are able to successfully issue all the lock requests, the system will delete the entire content of the lock container.

· If one of the locks in a container cannot be set, the function module FLUSH_ENQUEUE triggers the exception FOREIGN_LOCK.  In this case, none of the locks registered in the container will be set and the container content will remain complete for further dispatch attempts.

· You can delete the contents of the lock container using the function module RESET_ENQUEUE.

· If you set a lock when you call the respective lock module, you can specify the lock mode, which determines the type and purpose of the lock.  If you make no specification, the default in the definition of the respective lock object applies.

· The three different lock modes are E, X, and S:

· Lock mode E: This sets a lock for changing data. This lock can be accumulated (see following graphics).
Example: You wish to book a flight. Once you have chosen the flight you want to book, you should make sure that no other customer books the same flight to prevent the last free seat from being occupied more than once. To do this, you must lock the respective flight (SFLIGHT entry) with mode E.

· Lock mode X: This mode is used like mode E for changing data. The only technical difference from mode E is that the respective lock does not allow accumulation (for details, see the following graphics).

· Lock mode S: This mode ensures that data displayed in your program cannot be changed by other users during the entire display time. Here you do not want to change the data yourself.
Example: Your program has determined a price for a flight and has displayed this price to an interested customer. While the customer is considering whether or not to book the flight, you want to ensure that the price will not change.

[image: image18.wmf]ã

SAP

AG 2002

Effect of

Lock

Modes

(

View

:

Other User

)

Mode

Table

Lock

argument

E         SFLIGHT

800LH 040019990101

X         SFLIGHT

800LH 040219990131

S         SFLIGHT

800LH 240219991231

E , X , S

S

E , X

Program

(

User

B)

Existing locks

in a

program

(

User

A)

Attempt

to

set

a

lock

E , X , S


· If you have existing locks in the system, attempts by outsiders to set locks (lock request by the program of another user) in the same data record are treated as follows: 

· Existing exclusive locks (E or X) categorically reject every lock attempt of another user, irrespective of the mode in which the other user has attempted to set the lock.

· An existing shared lock (S), on the other hand, allows other shared locks for protected display to be set for one and the same data record.  Attempts by other users to set shared locks for the same data record will, of course, be rejected.  

· If you attempt to lock a data record more than once while a program is running (for example, using a function module that you call and this module itself sets locks), the lock system reacts in the following way:

· If you have an existing E lock, furhter E and S locks for the same data record will be accepted.  Only attempts to set an X lock for the same data record will be rejected. 

· If you have an existing X lock, every futher attempt to set a lock will be rejected.

· If you have an existing S lock, furhter S locks for the same data record can be set from within the same program.  If, in addition, there are no further shared locks set by another user for this data record, you can also set an additional E lock.   Naturally, no X lock can be set if there is an existing S lock.

· If you want to ensure that you are reading up-to-date data in your program (with the intention of changing the data and saving it to the database), perform the following steps in your program in the order given. These steps pertain to lock requests and database accesses:

· Setting locks for the data to be processed.

· If the lock has been successfully set, read the current data from the database.

· Change the program data (user inputs) and update the changes to the database.

· Release set locks once again.

· With this order of steps, you ensure not only that your changes run completely under the protection of locks, but also that you only read data hat has been changed consistently by other programs.  This assumes, however, that all the application programs use the SAP lock concept and adhere to the step sequence given. 

· If you do not adhere to the sequence "Lock -> Read -> Change -> Unlock", there is the danger that your program will read data from the database that is currently locked by another program and is also being processed (see slide).   In this case, this could also mean that, even if the lock is successfully set after the read action, the data read by your program and displayed to the user for change is already out of date.  This therefore makes it possible for a user of your program to enter changes for data that is no longer up to date.  For this reason, you should always follow the procedure described here.

Exercises

	Unit: SAP Locking Concept

Using the SAP Locking Concept

	At the conclusion of these exercises, you will be able to:

· Call and use lock modules.

· Locate the places in programs where locks must be set and released in order to ensure that the data to be changed is protected adequately against competing accesses. 

	Program SAPMZ##_BOOKINGS1 from the previous unit is to be changed to include locks that will prevent the booking data from being canceled and the flight data from being changed.

	Program:

SAPMZ##_BOOKINGS2
Transaction code: 
Z##_BOOKINGS2
Template:
SAPBC414T_BOOKINGS_02
Model solution:
SAPBC414S_BOOKINGS_02


1-1
Copy your solution SAPMZ##_BOOKINGS1 or the program template SAPBC414T_BOOKINGS_02 with all subobjects to SAPMZ##_BOOKINGS2 (## is your group number).  
Assign transaction code Z##_BOOKINGS2 to the program.

1-2
Call the lock modules ENQUEUE_ESFLIGHT, ENQUEUE_ESBOOK, ENQUEUE_ESFLIGHT_SBOOK and DEQUEUE_ALL in subroutines.  The subroutines in question are already created (blank) and combined in the Include MZ##_BOOKINGS2F02.  To supply the interface parameters of the lock modules with data, use the fields of the structures SDYN_CONN and SDYN_BOOK because these are placed in screens 100 and 301, respectively, as input fields.

1-3
Provide solutions for the exceptions of the lock modules.
Possible user messages:

Data record is already being edited

(
Message 060

Processing terminated (booking already locked)
(
Message 061

Flight and/or bookings are already being edited
(
Message 062

Lock request not successful

(
Message 063

1-4
Protect the database changes related to the booking cancelations by calling the corresponding lock modules (by calling the corresponding subroutines). If a user action calls up screen 100, the locks must be canceled.

	[image: image27.wmf]
	The lock module ENQUEUE_ESFLIGHT enables locks to be set for entries in table SFLIGHT. The lock module ENQUEUE_ESBOOK  enables locks to be set for entries in table SBOOK.  The lock module ENQUEUE_ESFLIGHT_SBOOK enables locks to be set in both tables at the same time (SFLIGHT, SBOOK). This allows you to lock a flight with all the bookings belonging to it..


Optional Exercise

2
Extend your program for creating a new customer to include the necessary lock module calls.  The calls ENQUEUE_ESCUSTOM (lock customer) and DEQUEUE_ALL (remove all locks) are already coded and encapsulated in the subroutines ENQ_SCUSTOM and DEQ_ALL (Include BC414T_CREATE_CUSTOMER_02F01).

2-1
Copy your solution SAPMZ##_CUSTOMER1 or the program template SAPBC414T_CREATE_CUSTOMER_02 with all subobjects to SAPMZ##_CUSTOMER2 (## is your group number).  Assign transaction code Z##_CUSTOMER2 to the program.

2-2
Insert the call for the subroutines ENQ_SCUSTOM and DEQ_SCUSTOM at the appropriate places in your program.  When should the customer data record be locked?  Locate all the places where the data record lock must be canceled.  Familiarize yourself with the program flow, using the debugger if necessary.

	Unit: SAP Locking Concept

Lock Objects (Optional)

	At the conclusion of this exercise, you will be able to:

· Search for and find lock objects.




3-1
Find out which function modules are maintained for logically locking flights, bookings, and flights with all dependent bookings in the system.

Solutions

	Unit: SAP Locking Concept

Topic: Using the SAP Locking Concept


Model Solution SAPBC414S_BOOKINGS_02

1 PAI Modules

*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_02I01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'CANCEL'.
      CASE sy-dynnr.
        WHEN '0100'.
          LEAVE PROGRAM.
        WHEN '0200'.
* remove all database locks
          PERFORM deq_all.
          LEAVE TO SCREEN '0100'.
        WHEN '0300'.
          LEAVE TO SCREEN '0100'.
        WHEN OTHERS.
      ENDCASE.
    WHEN 'EXIT'.
      LEAVE PROGRAM.
    WHEN OTHERS.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
****************************CANCEL BOOKING**************************
    WHEN 'BOOKC'.
* set database lock for selected flight and depending bookings
      PERFORM enq_sflight_sbook.
      PERFORM read_sflight USING wa_sflight sysubrc.
      PERFORM process_sysubrc_bookc.
      PERFORM read_spfli USING wa_spfli.
      PERFORM read_sbook USING itab_book itab_cd.
      REFRESH CONTROL 'TC_SBOOK' FROM SCREEN '0200'.
****************************CREATE BOOKING**************************
    WHEN 'BOOKN'.
      PERFORM read_sflight USING wa_sflight sysubrc.
      PERFORM process_sysubrc_bookn.
      PERFORM read_spfli USING wa_spfli.
      PERFORM initialize_sbook USING wa_sbook.
    WHEN 'BACK'.
      SET SCREEN 0.
    WHEN OTHERS.
      SET SCREEN '0100'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0200 INPUT.
  CASE save_ok.
    WHEN 'SAVE'.
      PERFORM collect_modified_data USING itab_sbook_modify.
      PERFORM save_modified_booking.
* remove all database locks
      PERFORM deq_all.
      SET SCREEN '0100'.
    WHEN 'BACK'.
* remove all database locks
      PERFORM deq_all.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0200'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0200  INPUT
FORM Routines

1.2 F01
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_02F01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  PROCESS_SYSUBRC_BOOKC
*&-----------------------------------------------------------------*

FORM process_sysubrc_bookc.
  CASE sysubrc.
    WHEN 0.
      SET SCREEN '0200'.
    WHEN OTHERS.
* remove all database locks
      PERFORM deq_all.
      MESSAGE e023 WITH sdyn_conn-carrid sdyn_conn-connid
                        sdyn_conn-fldate.
  ENDCASE.
ENDFORM.                               " PROCESS_SYSUBRC_BOOKC
F02
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_02F02
*------------------------------------------------------------------*

*------------------------------------------------------------------*

*   FORM ENQ_SFLIGHT
*------------------------------------------------------------------*

FORM enq_sflight.
  CALL FUNCTION 'ENQUEUE_ESFLIGHT'
       EXPORTING
            carrid         = sdyn_conn-carrid
            connid         = sdyn_conn-connid
            fldate         = sdyn_conn-fldate
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e060.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               "ENQ_SFLIGHT
*------------------------------------------------------------------*

*   FORM ENQ_SBOOK
*------------------------------------------------------------------*

FORM enq_sbook.
  CALL FUNCTION 'ENQUEUE_ESBOOK'
       EXPORTING
            carrid         = sdyn_book-carrid
            connid         = sdyn_book-connid
            fldate         = sdyn_book-fldate
            bookid         = sdyn_book-bookid
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e061.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               "ENQ_SBOOK
*------------------------------------------------------------------*

*   FORM ENQ_SFLIGHT_SBOOK
*------------------------------------------------------------------*

FORM enq_sflight_sbook.
  CALL FUNCTION 'ENQUEUE_ESFLIGHT_SBOOK'
       EXPORTING
            carrid         = sdyn_conn-carrid
            connid         = sdyn_conn-connid
            fldate         = sdyn_conn-fldate
       EXCEPTIONS
            foreign_lock   = 1
            system_failure = 2
            OTHERS         = 3.
  CASE sy-subrc.
    WHEN 0.
    WHEN 1.
      MESSAGE e062.
    WHEN OTHERS.
      MESSAGE e063 WITH sy-subrc.
  ENDCASE.
ENDFORM.                               "ENQ_SFLIGHT_SBOOK
*-------------------------------------------------------------------

*   FORM DEQ_ALL
*------------------------------------------------------------------*

FORM deq_all.
  CALL FUNCTION 'DEQUEUE_ALL'.
ENDFORM.                               "DEQ_ALL
F03
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_02F03
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  READ_SPFLI
*&-----------------------------------------------------------------*

*      -->P_WA_SPFLI  text
*----------------------------------------------------------------------*

FORM read_spfli USING p_wa_spfli TYPE spfli.
  SELECT SINGLE * FROM spfli INTO p_wa_spfli
         WHERE carrid = sdyn_conn-carrid
         AND   connid = sdyn_conn-connid.
  IF sy-subrc <> 0.
* remove all database locks
    PERFORM deq_all.
    MESSAGE e022 WITH sdyn_conn-carrid sdyn_conn-connid.
  ENDIF.
ENDFORM.                               " READ_SPFLI
OPTIONAL:

Model Solution SAPBC414S_CREATE_CUSTOMER_02

2 FORM Routines

3 F01
*------------------------------------------------------------------*

***INCLUDE BC414S_CREATE_CUSTOMER_02F01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  SAVE
*&-----------------------------------------------------------------*

FORM save.
PERFORM number_get_next USING scustom.
* lock dataset
  PERFORM enq_scustom.
PERFORM save_scustom.
* unlock dataset
  PERFORM deq_all.
ENDFORM.    " SAVE

